

LABORATOR 4

Analiza de curent alternativ. Filtre analogice

Cuprins

1 Obiectivul lucrării	2
2 Analiza de curent alternativ folosind Spice. Filtre analogice	2
2.1 Ce este analiza de curent alternativ?	2
2.2 Circuit R-L serie	4
2.3 Circuit R-C serie	6
2.4 Circuit R-L-C serie	6
2.5 Transferul maxim de putere	9
3 Exemple pentru domenii de frecvențe	10
3.1 Rezonanța RLC serie	10
3.2 Filtre pasive	17
3.2.1 Filtrul trece-jos și filtrul trece-sus	17
3.2.2 Filtrul trece-bandă de tip Butterworth	19
3.2.3 Filtrul oprește-bandă	20
4 Probleme	21
5 Concluzii	22
6 Referat de laborator	22

1 Obiectivul lucrării

În această lucrare vom analiza circuite în regim permanent sinusoidal (curent alternativ) - deci cu frecvență fixă dar și circuite ce iau în calcul un domeniu de frecvențe, de exemplu filtrele analogice. După o prezentare teoretică a analizei AC din Spice, sunt prezentate câteva exemple simple de circuite la care se folosește această analiză. La final sunt propuse câteva probleme pe care trebuie să le rezolvați voi.

2 Analiza de curent alternativ folosind Spice. Filtre analogice

2.1 Ce este analiza de curent alternativ?

Spice poate fi folosit pentru analiza de curent alternativ - AC (fie că este vorba de o singură frecvență, fie că este vorba de un domeniu de frecvențe). Pentru obținerea rezultatelor în cadrul analizei de curent continuu am avut nevoie de tandemul comezilor ".DC" și ".PRINT DC". Acum avem o situație similară folosind comenzile ".AC" și ".PRINT AC".

Spice calculează în modul de simulare de curent alternativ *răspunsul în frecvență al circuitelor liniare și al celor echivalente de semnal mic*. Spice folosește pentru analiza de curent alternativ metoda nodală modificată în complex și operează cu fazori.

Până acum toate sursele de tensiune și de curent folosite erau în curent continuu. Sintaxa surselor AC este asemănătoare.

Informațiile ce trebuiesc furnizate sunt:

- numele sursei care trebuie să înceapă cu V sau I;
- nodurile între care este conectată;
- amplitudinea (valoarea maximă) (X_{\max});
- faza sa inițială (α).

Separat, frecvența *tuturor* surselor din circuit este definită printr-o comandă ".AC".

Exemple de definiții de surse de curent alternativ:

```
*nume lista nodurilor tip valoare fază(grade)
V1ac 1 0 AC 230V 45
Vba 2 3 AC 240; fază inițială 0 grade
I3y 3 4 AC 10.0A -45; fază inițială -45 grade
Is1 1 9 AC 35mA; 35 miliamperi la 0 grade
```

Specificarea tipului AC este obligatorie la definirea surselor deoarece cel implicit este DC. Faza este măsurată în grade. Dacă nu se precizează faza, aceasta se presupune 0. În exemplele anterioare V și A (ca unitați de măsură) sunt opționale după valorile tensiunii și ale curentului.

Rezistorul se declară la fel ca în curent continuu:

R_ nume N1 N2 valoare_in_Ohm

Condensatorul liniar se declară în Spice astfel:

C_ nume N1 N2 valoare_in_Farad

Bobina liniară se declară în Spice astfel:

L_ nume N1 N2 valoare_in_Henri

Pentru a putea folosi comanda **.PRINT AC**, trebuie activată comanda **.AC**.

Comanda **.AC** este folosită pentru a parcurge un domeniu de frecvențe pentru un circuit dat. Aceasta corespunde găsirii răspunsului în frecvență. Există 3 tipuri de parcurgere a domeniului frecvență: **LIN**, **DEC** și **OCT**.

Exemple de comenzi **.AC**:

```
.AC LIN 1 50Hz 50Hz  
.AC LIN 11 150 250; o baleiere liniară a domeniului  
.AC DEC 20 10Hz 100kHz; o baleiere pe decade a domeniului
```

Prima comandă realizează o analiză numai pentru frecvența de 50Hz (am ales domeniul liniar). Adăugarea unităților Hz după valoare este opțională. Cea de-a doua comandă parcurge domeniul format din frecvențele 150Hz, 160Hz, 170Hz, 180Hz, 190Hz, 200Hz, 210Hz, 220Hz, 230Hz, 240Hz și 250Hz. Cea de-a treia comandă parcurge logaritmice domeniul folosind 20 de puncte pe decadă pe un domeniu de 4 decade (10Hz-100Hz, 100Hz-1kHz, 1kHz-10kHz și 10kHz-100kHz).

Să analizăm acum comanda **.PRINT AC**. Afișarea componentelor fazoriale (numere complexe) necesită anumite opțiuni. Există 4 expresii necesare pentru aceasta: modul, fază, parte reală și parte imaginară. Pot fi afișate tensiuni și curenți. De exemplu, pentru a afișa modulul unei tensiuni dintre nodurile 2 și 3, vom specifica **VM(2,3)**, sau simplu **V(2,3)**. Faza în grade pentru aceeași tensiune va fi **VP(2,3)**. Dacă dorim modulul curentului prin rezistorul $R_{sarcina}$, se specifică **IM($R_{sarcina}$)**. Partea reală a tensiunii la nodul 7 va fi **VR(7)** iar partea sa imaginară, **VI(7)**. Ca și în cazul comenzii **.PRINT DC**, nu există un număr limită de câte ori poate fi folosită.

Exemple:

```
.PRINT AC VM(30,9) VP(30,9); modulul și faza tensiunii  
.PRINT AC IR(Rs) II(Rs); părțile reală și imaginară ale curentului prin Rs  
.PRINT AC VM(13) VP(13) VR(13) VI(13); toate informațiile la nodul 13
```

Pentru afișarea unor grafice în funcție de frecvență, se folosește comanda **.PROBE**. Pentru a vizualiza diferite variații, folosim din meniu *Trace->Add Trace* de unde putem alege fie mărimea respectivă, fie diferite funcții ale mărimii. (De exemplu pentru variația fazelor, se alege funcția **P()**, pentru variația părții imaginare, se alege funcția **IMG()**, pentru variația părții reale, se alege funcția **R()**.)

2.2 Circuit R-L serie

În figura 1 este prezentat un circuit format dintr-o rezistență $R_1=10\Omega$ și o bobină $L_1=10\text{mH}$, legate în serie. Combinația poate reprezenta, de exemplu, cazul unei bobine reale. Vom calcula curentul prin circuit pentru un semnal sinusoidal aplicat la intrare V_{in} , cu valoarea maximă 1V și frecvență $f=100\text{Hz}$.




Fig. 1. Circuit R-L serie

Se folosește următorul program:

```
Circuit R-L serie
Vin 1 0 AC 1
R1 1 2 10
L1 2 0 10m
.AC LIN 1 100 100
.PRINT AC I(R1) IP(R1) IR(R1) II(R1)
.END
```

În urma rulării programului se obține fișierul de ieșire .out:

```
**** 04/01/21 17:10:46 ***** Evaluation PSpice (Nov 1999) *****

Circuit R-L serie
****      CIRCUIT DESCRIPTION
*****
*
Vin 1 0 AC 1
R1 1 2 10
L1 2 0 10m
.AC LIN 1 100 100
.PRINT AC I(R1) IP(R1) IR(R1) II(R1)
.END

**** 04/01/21 17:10:46 ***** Evaluation PSpice (Nov 1999) *****

Circuit R-L serie
```

```

****      SMALL SIGNAL BIAS SOLUTION      TEMPERATURE =  27.000 DEG C

*****
*

      NODE      VOLTAGE      NODE      VOLTAGE      NODE      VOLTAGE      NODE      VOLTAGE

(   1)    0.0000  (   2)    0.0000

      VOLTAGE SOURCE CURRENTS
      NAME          CURRENT
      Vin            0.000E+00
      TOTAL POWER DISSIPATION  0.00E+00  WATTS

**** 04/01/21 17:10:46 ***** Evaluation PSpice (Nov 1999) *****

Circuit R-L serie
****      AC ANALYSIS      TEMPERATURE =  27.000 DEG C
*****
*

      FREQ          I(R1)          IP(R1)          IR(R1)          II(R1)

      1.000E+02      8.467E-02      -3.214E+01      7.170E-02      -4.505E-02

      JOB CONCLUDED
      TOTAL JOB TIME          .03

```

Observații:

Calculul punctului de polarizare are ca rezultat valori nule pentru toate potențialele nodurilor deoarece în circuit nu avem surse de curent continuu.

Curentul prin sursă și puterea debitată de ea sunt nule din aceleași cauze.

Frecvența de analiză este de 100Hz, amplitudinea curentului prin circuit este de 84mA și este defazat în urma tensiunii cu 32°.

Aceste rezultate permit calcularea următoarelor mărimi:

Impedanța circuitului:

$$Z = V_{in}/I(R_1) = 11.9 * \exp(j * 32^\circ)$$

Puterile activă, reactivă și aparentă:

$$\begin{aligned}
 P &= U * I * \cos\varphi = 1V * 0.084A * \cos 32^\circ = 0.0712W \\
 Q &= U * I * \sin\varphi = 1V * 0.084A * \sin 32^\circ = 0.0445VAR \\
 S &= U * I = 1V * 0.084A = 0.084VA
 \end{aligned}$$

2.3 Circuit R-C serie

Circuitul R-C serie cu valorile $R_1=10\Omega$ și $C_1=100\mu\text{F}$ este prezentat în figura 2. Vom analiza circuitul la frecvență $f=318\text{Hz}$.




Fig. 2. Circuit R-C serie

```
Circuit R-C serie
Vin 1 0 AC 1
R1 2 0 10
C1 1 2 100u
.AC LIN 1 318 318
.PRINT AC I(R1) IP(R1) V(R1) V(C1)
.END
```

Folosind fișierul de ieșire arătați că $V_{in} = \sqrt{V^2(R_1) + V^2(C_1)}$

2.4 Circuit R-L-C serie

Circuitul R-L-C serie este prezentat în figura 3. Comportarea sa în regim sinusoidal depinde de frecvență și putem distinge trei situații.




Fig. 3. Circuit R-L-C serie

a) Rezonanța tensiunilor (comportare rezistivă)

Aceasta are loc atunci când $U_L=U_C$ deci $X_L=X_C$. Aceasta se întâmplă la frecvența:

$$\omega = \omega_0 = \frac{1}{\sqrt{LC}}$$

La această frecvență circuitul are un caracter pur rezistiv, impedanța sa fiind minimă:

$$|Z| = |Z|_{\min} = \sqrt{R^2 + (X_L - X_C)^2} = R$$

iar curentul prin circuit are valoarea maximă egală cu:

$$I_{\max} = U/Z_{\min} = U/R$$

b) Comportare capacitivă

Pentru $\omega = \omega_1 < \omega_0$, avem $X_C > X_L$ și deci $U_C > U_L$ iar circuitul se va comporta ca un grup R-C echivalent și va determina un defazaj al curentului înaintea tensiunii la borne.

c) Comportare inductivă

Pentru $\omega = \omega_2 > \omega_0$, avem $X_L > X_C$ și deci $U_L > U_C$ iar circuitul se va comporta ca un grup serie R-L echivalent și va determina un defazaj al curentului în urma tensiunii la borne.

Considerând valorile $R=100\Omega$, $L=25.33\text{mH}$ și $C=1\mu\text{F}$, frecvența de rezonanță are aproximativ valoarea $f_0=1000\text{Hz}$. Rulați următorul fișier Spice:

```
Circuit R-L-C serie
Iin 0 1 AC 1
R 1 2 100
L 2 3 25.33m
C 3 0 1u
RC 3 0 1meg
.AC LIN 1 1000 1000
.PRINT AC V(1) VP(1)
.PRINT AC V(R) VP(R) V(L) VP(L) V(C) VP(C)
.END
```

Linia de program RC 3 0 1meg a fost introdusă deoarece sursa de curent are rezistență internă infinită și astfel nodurile 1, 2 și 3 nu ar avea o cale spre masă (pentru a vă convinge încercați varianta fără R_C). Rezistența $R_C=1\text{M}\Omega$ introdusă în paralel cu condensatorul asigură această cale de curent fără a modifica practic funcționarea circuitului.

```
Circuit R-L-C serie

****      AC ANALYSIS                      TEMPERATURE =  27.000 DEG C

*****
*
FREQ          V(1)          VP(1)

1.000E+03    1.000E+02    -1.063E-03
```

```

**** 03/30/21 17:45:18 ***** Evaluation PSpice (Nov 1999) *****

Circuit R-L-C serie

****      AC ANALYSIS                      TEMPERATURE =  27.000 DEG C

*****
*

FREQ      V(R)      VP(R)      V(L)      VP(L)      V(C)

1.000E+03  1.000E+02  -1.242E-19  1.592E+02  9.000E+01  1.592E+02

**** 03/30/21 17:45:18 ***** Evaluation PSpice (Nov 1999) *****

Circuit R-L-C serie

****      AC ANALYSIS                      TEMPERATURE =  27.000 DEG C

*****
*

FREQ      VP(C)
1.000E+03  -8.999E+01

JOB CONCLUDED

TOTAL JOB TIME      0.00

```

Rulați analiza de mai sus și pentru $f_1=500\text{Hz}$ și $f_2=2000\text{Hz}$ și comparați rezultatele.

Tot ca exercițiu schimbați alimentarea într-o sursă de tensiune sinusoidală de amplitudine 1V. Observați că în acest caz rezistența R_C în paralel cu C nu mai este necesară. De ce?

2.5 Transferul maxim de putere

Transferul maxim de putere în curent continuu are loc pentru cazul în care rezistența de sarcină este egală cu rezistența internă a sursei echivalente de tensiune. În cazul circuitelor de curent alternativ, atât impedanța internă a sursei echivalente cât și impedanța de sarcină sunt mărimi complexe. Se poate demonstra că într-o astfel de situație, transferul maxim de putere are loc dacă impedanța de sarcină și impedanța internă a sursei sunt complex conjugate.




Fig. 4. Circuit cu transfer maxim

Pentru circuitul din figura 4 de exemplu V_{in} are amplitudinea 10V și frecvența 1kHz iar $R_i=10\Omega$ și $L_i=25.33\text{mH}$. Atunci pentru ca transferul să fie maxim, trebuie ca $R_L=10\Omega$ și $C_L=1\mu\text{F}$. De ce? Rulați următorul exemplu:

```
Transfer maxim de putere
Vin 1 0 AC 10
RI 1 2 10
LI 2 3 25.33M
RL 3 4 10
CL 4 0 1U
.AC LIN 1 1K 1K
.PRINT AC I(RL) IP(RL)
.END
```

Se observă că $I(R_L)=0.5\text{A}$ ceea ce conduce la o putere absorbită egală cu 2.5W.

3 Exemple pentru domenii de frecvențe

3.1 Rezonanța RLC serie

Să analizăm circuitul RLC serie din figura 5. În acest caz rezonanța tensiunilor are loc la frecvența:

$$f_0 = \frac{1}{2\pi\sqrt{LC}} = 999.3\text{Hz}$$





Fig. 5. Circuit R-L-C serie

Vom analiza comportarea circuitului în gama de frecvențe cuprinse între 100Hz și 10kHz. Rulați următorul program și urmați pașii de mai jos:

```
Rezonanta serie
Vin 1 0 AC 1
R 1 2 100
L 2 3 25.33m
C 3 0 1u
.AC LIN 1000 100Hz 10kHz
.PROBE
.END
```


RULAȚI PROGRAMUL:




SE DESCHIDE EDITORUL PROBE:




SELECTAȚI AXIS SETTINGS




SELECTAȚI LOG CA ÎN FIGURĂ




SELECTAȚI ADD TRACE




SELECTAȚI CA ÎN FIGURĂ




SE OBSERVĂ MAXIMUL LUI $I(R)$ LA 1000HZ




VOM ADĂUGA GRAFICUL FAZEI CURENTULUI




APĂSAȚI ADD TRACE




INTRODUCEȚI FAZA CURENTULUI




SE OBȚIN GRAFICELE CA ÎN FIGURĂ




APĂSAȚI ADD Y AXIS



ADĂUGAȚI IMPEDANȚA $Z(\Omega) = V(1)/I(R)$



PENTRU A VEDEA CURSORUL, APĂSAȚI CA ÎN FIGURĂ



Activând cursorul ca în ultima imagine de mai sus, stabiliți valoarea impedanței ($Z(\omega) = \frac{V(I)}{I(R)}$) și a fazei pentru 3 frecvențe: $f_1=500\text{Hz}$, $f_0=1000\text{Hz}$ și $f_2=2000\text{Hz}$. (Obs: cursorul A1 se modifică cu butonul din stânga al mouseului iar cursorul A2 cu butonul din dreapta).

Se remarcă că pentru $f < f_0$, circuitul se comportă capacitiv, iar pentru $f > f_0$, el se comportă inductiv.


3.2 Filtre pasive

Filtrele pasive, conținând numai rezistoare, bobine și condensatoare, își găsesc numeroase aplicații în practică. Un filtru analogic este de obicei un cuadripol (sau diport) cu o poartă de intrare și o poartă de ieșire. Caracteristicile filtrului (de modul și de fază) conduc la transformarea semnalului de intrare în semnal de ieșire. Distingem filtre trece jos (FTJ) care lasă să treacă doar componentele de frecvență mică, filtre trece sus (FTS) care lasă să treacă doar componentele de frecvență înaltă, filtre trece bandă (FTB), filtre oprește bandă (FOB), filtre pieptene (lasă să treacă doar anumite frecvențe).

Cele mai simple filtre se obțin prin înserierea unui rezistor cu un condensator.

3.2.1 Filtrul trece-jos și filtrul trece-sus

În figura de mai jos avem de fapt un filtru trece jos dacă considerăm drept ieșire tensiunea pe condensator și un filtru trece sus dacă considerăm ieșirea tensiunea pe rezistor.



Folosim următorul program Spice:

```
Filtru trece-jos
Vin 1 0 AC 1
R 2 0 10k
C 1 2 10nF
.ac dec 10 1hz 1meghz
.probe
.end
```




Fig. 6. Determinarea benzii de trecere a FTJ

Trasați și dreapta de valoare = 0.707 și notați frecvența de intersecție cu graficul. Aceasta se numește frecvență de tăiere. (Obs: cursorul A1 se modifică cu butonul din stânga al mouseului iar cursorul A2 cu butonul din dreapta)




Fig. 7. Frecvența de tăiere

Pe un alt grafic reprezentați faza tensiunii de pe condensator, VP(1,2). Verificați cu ajutorul celui alt cursor că, la frecvența de tăiere faza este de -45° .

Ca un exercițiu, vă propunem analizarea într-un mod asemănător și a comportării filtrului trece-sus (ieșirea pe rezistor).

3.2.2 Filtrul trece-bandă de tip Butterworth

Schema unui filtru trece-bandă de tip Butterworth este prezentată în figura 8.




Fig. 8. Filtrul trece-bandă de tip Butterworth

Se folosește următorul program:

```
Filtru trece-banda
*de tip Butterworth
Vin 1 0 ac 1
l12 1 2 100u
c23 2 3 2u
c30 3 0 50u
l30 3 0 4u
r30 3 0 1
.ac dec 1000 5k 25k
.probe
.end
```

După rulare afișați curba V(3). Banda de trecere se determină citind coordonatele punctelor de intersecție cu dreapta 0.707. Se obține B=2.252kHz.




Fig. 9. Banda de trecere a filtrului trece-bandă

3.2.3 Filtrul oprește-bandă

Schema unui filtru oprește-bandă este prezentată în figura 10.




Fig. 10. Filtru oprește-bandă

Se folosește următorul program:

Filtru oprește-banda

Vin 1 0 AC 1

R12 1 2 10

R23 2 3 0.5

C34 3 4 26.5U

L40 4 0 265M

R20 2 0 10

.ac OCT 100 40 80

.probe

.end

După rulare afișați curba V(2). Frecvențele de tăiere se determină citind coordonatele punctelor de intersecție cu dreapta $0.707 \cdot 500\text{mV}$ deoarece amplitudinea în benzile de trecere este de 500mV . Se obțin $f_1=58.4\text{Hz}$ și $f_2=61.7\text{Hz}$.




Fig. 11. Filtrul oprește-bandă

4 Probleme

Găsiți curenții (modul și fază) prin laturile următoarelor circuite folosind PSpice (pentru toate circuitele frecvența este $f=50\text{Hz}$):

Problema 1.1

Se dau: $L_1 = \frac{100}{\pi} \text{ mH}$, $L_2 = \frac{300}{\pi} \text{ mH}$, $C = \frac{500}{\pi} \mu\text{F}$, $R=10\Omega$, $V_1 = 20\sin(\omega t - \pi/4)[\text{V}]$, $V_2 = 60\sin(\omega t + \pi/4)[\text{V}]$






Fig. 12. Problema 1.1

Problema 1.2

Se dau: $L_1 = \frac{200}{\pi} \text{ mH}$, $L_2 = \frac{100}{\pi} \text{ mH}$, $C = \frac{1000}{\pi} \mu\text{F}$, $R = 10 \Omega$, $V_1 = 20\sqrt{2}\sin(\omega t - \pi/2) [\text{V}]$,
 $V_2 = 20\sin(\omega t + 3\pi/4) [\text{V}]$

**Fig. 13.** Problema 1.2**Problema 1.3**

Se dau: $L = \frac{100}{\pi} \text{ mH}$, $C = \frac{250}{\pi} \mu\text{F}$, $R_1 = 20 \Omega$, $R_2 = 10 \Omega$, $R_3 = 20 \Omega$, $V_1 = 40\sin(\omega t + \pi/4) [\text{V}]$,
 $V_2 = 80\sin(\omega t + \pi/4) [\text{V}]$

**Fig. 14.** Problema 1.3**5 Concluzii**

În această lecție am analizat circuite analogice în funcție de frecvență. Am considerat atât cazul în care frecvența este unică, cât și cazul în care frecvența baleiează un interval.

În lecția următoare vom studia răspunsul în timp al circuitelor, adică regimul tranzitoriu (TRANsient analysis).

6 Referat de laborator

Veți încărca pe Moodle un fișier .pdf care să conțină codul .out, de la toate exercițiile, împreună cu figurile cu numerotarea nodurilor.